

Metamorfosi dei tipi sublacensi

Claudio Vincoletto

Sommario

L'accurato recupero di un classico, nella sua prima versione digitale: un carattere tipografico fondato sulla calligrafia tardomedievale e impiegato nei primi incunaboli italiani, quindi riproposto agli inizi del XX secolo dall'ultima delle grandi *private presses* inglesi.

Abstract

The first and accurate digital revival of a classic typeface, based on the medieval calligraphy and used for the first time in Italian incunables. A copy of this original was employed by the last representative of *private press movement*.

Fin da bambino provo un gusto innato per le cose strane e bizzarre, per la frequentazione delle zone liminari, varcandone poi spesso i confini allo scopo di inoltrarmi in quelle che gli antichi chiamavano *terrae monstruum*. Ricordo che, sfogliando i prediletti libri di storia naturale, rimanevo affascinato dal concetto di “anello di congiunzione”, che animava i dibattiti fra i postulati della paleontologia e le teorie legate all'evoluzione. Come spiegare nello svolgersi del tempo alcuni ritrovamenti fossili che si presentavano dal nulla, come innovatori incontrastati di ciò che sarebbe venuto, tanto da mettere in crisi l'idea stessa di genealogia?

Tutto questo mi ha spinto ad affrontare uno dei caratteri più celebrati e, al tempo stesso, più problematici della storia della tipografia. Risale al 1465 il primo esemplare attestato di un libro stampato in Italia, a Subiaco, nei pressi di Roma. Giunti appena un anno prima, due monaci tedeschi — Conrad Sweynheym e Arnold Pannartz — per la prima volta sperimentarono quanto appreso nelle officine di Fust e Schöffer (allievi e soci di Gutenberg), adattando la forma delle lettere a tipi di scrittura più diffusi nella penisola e ispirati ai calligrafi umanisti, creando una sorta di ibrido oggi denominato Gotico-Antiqua (BOARDLEY, 2019).

Diedero alle stampe solo quattro volumi con quel carattere, di cui solo tre sono pervenuti fino a noi (il *De oratore* di Cicerone, le *Opere* di Lattanzio, il *De civitate Dei* di Agostino). Trasferitisi a Roma nel 1467, e riallestendo totalmente la loro officina, ne forgiarono uno nuovo, comunemente ritenuto meno pregiato. Uno degli spunti primari del mio progetto venne fornito proprio da una copia del *De oratore* stampata a Subiaco, che è disponibile oggi


in una pregevole edizione anastatica accompagnata da un'esauriente introduzione e una dettagliata scheda bibliografica (CICERO, 2015).

Per comprendere l'originalità di Sweynheym e Pannartz, va tenuto presente che quello finora discusso è un periodo che vede non solo lo sviluppo tecnologico delle modalità di scrittura (da quella manoscritta a quella stampata), ma che coincide anche con un'innovazione essenzialmente stilistica (il passaggio dalla forma gotica delle lettere a quella tonda) di cui il carattere forgiato a Subiaco rappresenta la transizione. Alcuni studi individuano, in diversi manoscritti presenti nei monasteri di Santa Scolastica e del Sacro Speco, le fonti che ispirarono i due prototipografi (ICCU, 2019). Si ritiene specificamente che per le lettere minuscole, più calligrafiche, il riferimento sia il ms. XXXIV, mentre le maiuscole sono riconducibili in modo più generico alle epigrafi in caratteri capitali e ai capilettera miniatati presenti in alcuni manoscritti della biblioteca, come per esempio il ms. CLX (DE GREGORI, 1942).

Nel considerare la trasposizione dal modello calligrafico ai tipi in metallo, propongo sia lecito ritenere che il disegno dei glifi vero e proprio sia stato progettato in termini geometrici. In questa direzione, con l'ausilio di METAFONT, ho tentato di immaginare quali schemi potessero aver seguito i due monaci, convinto che la rappresentazione dello spazio bidimensionale della scrittura, così nel carattere tipografico come nella font digitale, non si potesse discostare di molto dalla trattatistica coeva dedicata alle forme dei caratteri, fortemente impregnata di modelli matematici. Tra i primi esempi risalta il lavoro di Felice Feliciano (*Alphabetum Romanum*), che precede di qualche anno le stampe di Subiaco. Tuttavia questi studi, così come quelli di Pacioli e Tory, si concentrano sulle lettere capitali romane, mentre personalmente ho trovato più utile un'opera di poco più tarda in cui vengono affrontati con questi criteri anche i caratteri minuscoli.

Il trattato di Sigismondo Fanti è infatti dedicato in buona parte alla conversione dei tratti calligrafici in coordinate geometriche (FANTI, 2013). Su questo modello, con pochissimi aggiustamenti, ho riscontrato delle ricorrenze nel carattere Subiaco che mi hanno convinto a fissare dei precisi rapporti a partire dalle dimensioni del corpo del carattere (in 6 mm), suddiviso in 21 unità (*hair*). Ho programmato dunque METAFONT in modo che vengano calcolate tutte le misure verticali di riferimento. Il medesimo principio viene poi adoperato anche

Type 1:120R · Subiaco 1465



Riproduzione in digitale




Figura 1: Confronto con i glifi originali (TW, 2019).

per la componente orizzontale, glifo per glifo:

```
mode_setup;

%size#:=6mm#;
size#:=17pt#;

font_size size#;
font_identifier:="Subiaco";
font_coding_scheme:"TeX text";

% Proportions
hair# = 1/21 size#;
body_height# + body_depth# = size#;
body_height# = 14/21 size#;
x_height# = 8/21 size#;
asc_height# = 13.5/21 size#;
cap_height# = 13/21 size#;
bar_height# = 6.5/21 size#;
desc_depth# = 6/21 size#;
```

Non mi è possibile dimostrare che Sweynheym e Pannartz abbiano impiegato deliberatamente la successione di Fibonacci nelle proporzioni del disegno. Tuttavia, la suggestione è molto forte e può essere ricondotta alla diffusa mentalità condivisa dai sapienti dell'epoca. I recenti studi di Frank Blokland vanno in questa direzione, come ho avuto modo di accennare diversi anni fa, sebbene la sua ricerca non fosse conclusa (VINCOLETTA, 2013). Egli ipotizza che agli inizi della stampa vi sia stata l'esigenza di una normalizzazione dei tipi gotici, poi convertita anche nella fabbricazione dei tipi romani. Questo sarebbe evidente nelle relazioni morfologiche che sottostanno alla struttura di tali alfabeti. Pertanto, le convenzioni tipografiche si sarebbero sviluppate a partire da schemi ricorrenti, costruiti su modelli armonico-ritmici, completamente estranei a preferenze ottiche. A suo avviso, sarebbe in questo modo possibile ottenere maggiori informazioni sul processo creativo dei caratteri antichi, se non addirittura individuare mo-

delli parametrici per la progettazione di caratteri digitali (BLOKLAND, 2016).

L'utilizzo di METAFONT si rivela molto efficace nel costruire i glifi seguendo queste indicazioni, proprio in virtù del suo carattere essenzialmente parametrico. Tuttavia, per rendere alcuni aspetti della componente calligrafica del Subiaco ho dovuto modificare alcuni comportamenti del programma, forzandolo in due direzioni. Una delle mie priorità consisteva nell'elaborare il disegno usando esclusivamente dei contorni; l'altra, nel recuperare in modo obliquo l'idea delle penne virtuali proposta da Knuth. Il metodo adoperato si basa sul concetto di "interpolazione", per cui una volta fissata una serie di punti estremi, che sintetizzano l'andamento di un tratto tracciato da una penna, questi vengono collegati fra loro in una linea ciclica mediante delle curve di Beziér. Per rendere evidente l'applicazione di questi passaggi, viene riportato per intero il codice che una volta compilato genera la lettera "b", illustrata nella figura 2.

```
"Subiaco letter b";
beginsubiacochar("b",9,asc_height#,0);
penpos1(curve,pa); penpos2(curve,pa);
penpos3(curve,pa); penpos4(.9curve,pa);
penpos41(curve,pa); penpos12(curve,pa);
penpos23(curve,pa); penpos34(curve,pa);
penpos5(stem,pa); penpos6(stem,pa);
x1=x3=.5[x4,x2]; x2=w-1.25hair;
x4=x5=2hair; x6=1.9hair;
y1m=x_height#o; y3l=y6l=0-o;
y2=y4=.5[y3,y1]; y5=h;
p1=z1{right}..z2{down}..z3{left}..z4{up}..cycle;
z41= directionpoint dir pa of p1;
z12= directionpoint dir (pa-90) of p1;
z23= directionpoint dir (pa-180) of p1;
z34= directionpoint dir (pa-270) of p1;
curvestroke12(1,12,2,blob);
curvestroke23(2,23,3,1.5blob);
curvestroke34(3,34,4,blob);
z4'=whatever[z5,z6]; y4'=.6875x_height#;
```

```

numeric theta;
theta:=angle((x1,y1+.125x_height)-z4');
penpos4'(thin,theta-90);
penstroke56(5,6,.25,.48,.75,flx);
dish_serif(6,5,a,.7slab,jut,1.6jut);
p2=(dserif5 rotatedaround(z5,pa_f));
p3=z4'l{dir theta}..tension 3 and 1..{right}z1m;
p4=z1l{left}..tension 1 and 3..{-dir theta}z4'r;
p5=stroke56.r...z6r;
numeric i,j,g,f,r,s,t,u;
(t,u) = p3 intersectiontimes p5;
(s,r) = p4 intersectiontimes p5;
(i,j) = arc34.r intersectiontimes p5;
fill stroke56l...p2...subpath(0,u-.3blob) of p5..
  subpath(t+.3blob,length p3) of p3 & arc12.r &
  arc23.r & subpath(0,1.5) of arc34.r...cycle;
unfill arc34.l & arc23.l & arc12.l &
  subpath(0,s-.5blob) of p4..cycle;
penlabels(range 1 thru 44,4');
endchar;

```

Si è detto quanto il Subiaco porti in sé la commistione di elementi tipografici (regolarità e uniformità del tratto) ed elementi calligrafici (inclinazioni, *ductus* della penna). Pertanto, in una prima fase il lavoro si è concentrato sull'individuazione, mediante sintesi, dei componenti costitutivi di ogni tratto, da cui attingere per assemblare lettera per lettera. Nella costruzione del glifo in figura, infatti, si possono identificare almeno due tratti essenziali: uno verticale, l'asta a sinistra, e l'altro curvo, che costituisce l'arco, altrimenti detto “pancia”, della lettera. In genere, per simulare un tratto calligrafico con METAFONT, si usa l'istruzione `draw`, ma in questo caso sono stati ridefiniti alcuni automatismi di base del programma, creando apposite definizioni (*macro*). Sia d'esempio la definizione `penstroke`, per determinare i contorni delle linee diritte, di cui si darà una breve descrizione:

```

vardef penstroke@#(suffix $,$$)
(expr start_point,mid_point,end_point,curve) =
  z@#h.r=point start_point of (z$r--z$$r);
  z@#m.r=point mid_point of (z$r--z$$r);
  z@#l.r=point end_point of (z$r--z$$r);
  z@#h.n=point start_point of (z$n--z$$n);
  z@#m.n=point mid_point of (z$n--z$$n);
  z@#l.n=point end_point of (z$n--z$$n);
  if y$=y$$:
    if y$l>y$r:
      y@#m.r:=y@#m.r+curve;
      y@#m.n:=y@#m.n-curve;
    else:
      y@#m.r:=y@#m.r-curve;
      y@#m.n:=y@#m.n+curve;
    fi
  else:
    if x$l<x$r:
      x@#m.r:=x@#m.r-curve;
      x@#m.n:=x@#m.n+curve;
    else:
      x@#m.r:=x@#m.r+curve;
      x@#m.n:=x@#m.n-curve;
    fi
  fi
  stroke[@#].r:=z@#h.r..z@#m.r{z$$r-z$r}..z@#l.r;
  stroke[@#].l:=z@#l.n..z@#m.n{z$n-z$$n}..z@#h.n;
  labels(@#h,@#m,@#l,@#m.r,@#m.n,@#h.r,
    @#h.n,@#l.r,@#l.n);
enddef;
path stroke[] .l, stroke[] .r;

```




Figura 2: Particolare della lettera *b*.

Per valutare l'inserimento dei parametri nella definizione `penstroke`, si vedano le prime due linee del codice riportato. Ogni tratto rettilineo va distinto per essere riconosciuto dal sistema ed è pertanto numerato con (`@#`). Seguono i punti estremi della linea, quello iniziale (`$`) e quello finale (`$$`). Quindi l'indicazione, in decimale rispetto alla lunghezza della linea, dell'effettivo punto di partenza del contorno (`start_point`), quello intermedio (`mid_point`) e quello di arrivo (`end_point`). Le ragioni che portano a individuare un punto intermedio sono dovute alle caratteristiche del tratto calligrafico, per cui questo risulta tendenzialmente più spesso alle estremità. Donald Knuth ne illustra un caso generico a p. 28 nel suo *The METAFONTbook* (KNUTH, 1986). Tale punto di flessione risulta anche congeniale in fase di stampa, in quanto rende l'asta meno bombata e maggiormente uniforme. Il divario dello spessore del punto intermedio rispetto alle estremità è indicato dal valore espresso da `curve`.

Nel caso riportato, in merito alla lettera *b*, le istruzioni impartite sono le seguenti (`flx` corrisponde a `2/1000` di `size`):

```
penstroke56(5,6,.25,.48,.75,flx);
```

Un approccio differente si può incontrare nella definizione `curvestroke`, in cui il contorno del tratto curvo è ricostruito per quadranti. Oltre ai punti ortogonali, viene automaticamente assegnato un punto intermedio sull'arco definito, in funzione dell'angolatura della penna. Anche in questo caso propongo l'analisi del codice impiegato:

```

vardef curvestroke@#(suffix $,$,$,$$(expr blob) =
numeric sect[] .t, sect[] .u;
  if y$>y$$$ :
    if x$<x$$$ :
% quadr 1
      arc[@#].r:=subpath(0,1-blob) of (z$.m{right}..
z$.m)..subpath(blob,1) of (z$.r..
z$.r{down});
      (sect@#.t,sect@#.u) = (z$.n{up}..z$.n)
intersectiontimes (z$.l..z$.l{left});
      arc[@#].l:=subpath(0,sect@#.t-blob) of
(z$.n{up}..z$.n)..
subpath(sect@#.u+blob,1) of (z$.l..
z$.l{left});
    else:
% quadr 4
      arc[@#].r:=subpath(0,1-blob) of (z$.r{down}..
z$.r)..subpath(blob,1) of (z$.l..
z$.l{left});
      (sect@#.t,sect@#.u) = (z$.m{right}..z$.m)
intersectiontimes (z$.n..z$.n{up});
      arc[@#].l:=subpath(0,sect@#.t-blob) of
(z$.m{right}..z$.m)..
subpath(sect@#.u+blob,1) of (z$.n..z$.n{up});
    fi
  else:
    if x$>x$$$ :
% quadr 3
      arc[@#].r:=subpath(0,1-blob) of (z$.l{left}..
z$.l)..subpath(blob,1) of (z$.n..
z$.n{up});
      (sect@#.t,sect@#.u) = (z$.r{down}..z$.r)
intersectiontimes (z$.m..z$.m{right});
      arc[@#].l:=subpath(0,sect@#.t-blob) of
(z$.r{down}..z$.r)..
subpath(sect@#.u+blob,1) of (z$.m..
z$.m{right});
    else:
% quadr 2
      arc[@#].r:=subpath(0,1-blob) of (z$.n{up}..
z$.n)..subpath(blob,1) of (z$.m..
z$.m{right});
      (sect@#.t,sect@#.u) = (z$.l{left}..z$.l)
intersectiontimes (z$.r..z$.r{down});
      arc[@#].l:=subpath(0,sect@#.t-blob) of
(z$.l{left}..
z$.l)..subpath(sect@#.u+blob,1) of
(z$.r..z$.r{down});
    fi
  fi
enddef;
path arc[] .l, arc[] .r;

```

Si può notare come, oltre all'identificatore (@#), vengano indicati il punto di partenza (\$), quello mediano (\$\$) e quello terminale (\$\$\$). Si aggiunge inoltre un valore al parametro `blob`, che esprime la distanza sottratta rispettivamente ai due percorsi secanti, a partire dal loro punto di intersezione, per rendere una curva più morbida e senza spigoli. In tal modo, la soluzione assegnata al primo quadrante della lettera “b” è il seguente:

```
curvestroke12(1,12,2,blob);
```

Potrebbero essere fornite ancora diverse indicazioni relative al codice della lettera “b”, ma è necessaria una digressione storica per discutere le ragioni di alcuni accorgimenti adottati.

Il recupero di caratteri del passato non è una concezione moderna. Nel periodo umanistico, e

successivamente con l'introduzione dei tipi romani, i riferimenti erano la scrittura carolina per il minuscolo e la capitale quadrata per il maiuscolo. Come già affermato, la grafia impiegata nei manoscritti venne studiata secondo una prospettiva geometrica, e questo in qualche modo contribuì a semplificare la conversione della scrittura in funzione dell'innovazione tecnologica portata dall'invenzione della stampa a caratteri mobili.

Alla fine del XIX secolo, il movimento Arts and Crafts, promosso da William Morris, segna una nuova fase di questo processo, con un pionieristico approccio al disegno dei caratteri mediante la fotografia. Stimolato dal gusto critico di John Ruskin, egli condivise il proposito di ricorrere al Medioevo come fonte di ispirazione, non tanto per uno spirito reazionario da contrapporre alla modernità portata dall'industrializzazione, quanto per arginare un diffuso decadimento dei valori estetici con il recupero e la riattualizzazione storica di un patrimonio perduto. Tra le arti applicate, la tipografia assunse un ruolo peculiare, e la Kelmscott Press fondata da Morris divenne un modello di perfezione formale, che venne imitato dalle piccole stamperie private inglesi nate a cavallo fra i due secoli. Oltre all'elevata qualità delle carte, degli inchiostri e dei torchi, Morris attinse dagli incunaboli la progettualità grafica, l'impostazione della pagina, le decorazioni e il disegno di alcuni tipi che fece riprodurre per le sue edizioni. Inizialmente fu sua l'idea di recuperare nel 1892 il carattere forgiato a Subiaco e riproporlo per la propria casa editrice, ma il progetto non vide mai la luce (COCKERELL, 1898).

Solamente nel 1902 venne realizzata una vera e propria riproduzione in metallo del carattere Subiaco. L'opera fu commissionata da Charles St John Hornby per la sua Ashendene Press, e messa a punto da un gruppo di persone che avevano strettamente lavorato con Morris: Emery Walker fornì degli ingrandimenti fotografici tratti dalle opere di Sweynheym e Pannartz, mentre Sidney Cockerell ne abbozzò il disegno. L'incisione venne portata a termine da Edward Prince, che già aveva intagliato tutti i caratteri utilizzati prima dalla Kelmscott Press e poi dalla Doves Press. Per le iniziali calligrafiche delle sue pubblicazioni, la Ashendene si avvalse dei più celebri artisti dell'epoca: Graily Hewitt, Edward Johnston ed Eric Gill. Questo per rimarcare la linea di continuità che legò i progetti editoriali di Kelmscott, Doves e Ashendene Press.

Il disegno del Subiaco di St John Hornby si presenta molto fedele se affiancato a quello del 1465, e questo dipende probabilmente dai principi usati per adattarlo (MCNEIL, 2017). In questo senso, è interessante prendere in considerazione alcuni saggi di Hermann Zapf, in cui viene affrontato il tema del revival, proprio discutendo le metodiche fotografiche adottate in quel periodo. In *Caratteri da stampa e libri* (1962) e nelle *Trasformazioni nel di-*

NSTITVENTI míchí. Q. frater eū sermonē
 referre & mandare huic tertio libro quē post an-
 thomí disputationē crassus habuisset : acerbā sane
 recordatio : ueterē animi curā molestiamq; reno-
 uauit. Nam illud immortalitate dignū ingenū : illa
 humanitas : illa uirtus. L. crassí morte exticta su-
 bito est : uixit diebus decem post eū diem q hoc et superiore
 libro cōtineñ. Vt enī romam rediit extremo scēicoꝝ ludoy
 die uehemēter cōmotus : ea oratione que ferebat habita esse
 ī cōcione a philippo quē dixisse constabat uidendū sibi esse
 aliud cōsiliū : illo senatu se rē.p. regere nō posse : mane idibꝯ
 septembris. et ille et senatus frequēs uocatu drusi in curiam
 uenit. Ibí cū drusus multa de philippo questus esset : retulit
 ad senatū de illo ipso quod in eū ordinē consul tam grauiter
 in cōcione esset iuectus. Hic ut sepe inter hoīes sapiētissimos
 constare uidi quāq; hoc crasso : cū aliquid accuratius dixis-
 set semp fere cōtingisset : ut nūq; dixisse melius putaret : tamē
 ommū consensu sic esse tum iudicatū audiui ceteros a crasso
 semp omnes illo autē die etiā ipm a sese supatū. Deplorauit
 enī casum atq; orbitatē senatus cuius ordinis a consule qui
 quasi parēs bonus aut tutor fidelis esse deberet : tamq; ab alí-
 quo nefario predone diriperet patrimonium dignitatis. Ne-
 q; uero inqt esse mirādum si tum suis consiliis rem.p. pflí-
 gasset. consiliū senatus rei.p. repudiaret. Hic cum homini et
 uehemēti et disertō et in p̄mis fortī ad resistendum philippo
 quasi quasdā uerboꝝ faces admouisset : nō tulit ille : et gra-
 uiter exarsit pignoribusq; ablatís crassū instituit coercere.

Figura 3: Simulazione di una pagina del *De Oratore*, Subiaco 1465 (BVMC, 2019).

segno dei caratteri in seguito all'evoluzione tecnica (1967), Zapf ripercorre la storia del libro stampato, da Gutenberg all'inizio del XX secolo, equiparando tre diversi caratteri, tutti ispirati alla medesima fonte, ossia i tipi di Nicolas Jenson (ZAPF, 1991). Il Golden Type della Kelmscott, risalente al 1890, rappresenta una pietra miliare della storia della tipografia, per i motivi spiegati innanzi, ma secondo Zapf è pure un fallimento. Gli ingrandimenti fotografici erano difettosi, forse per una carta bagnata eccessivamente, per cui il ricalco del contorno dei glifi risulta molto diverso dall'originale quattrocentesco. Morris ne era consapevole, tuttavia era interessato più all'effetto unitario dato dal colore scuro, quasi gotico, delle sue stampe, piuttosto che a un vero recupero filologico. Il Doves Type fu sviluppato tra il 1899 e il 1901 da Emery Walker e il suo assistente Percy Tiffin, su commissione di Thomas Cobden-Sanderson. Nuove e più accurate impressioni fotografiche consentirono di disegnare un carattere più leggero e maggiormente vicino a quello di Jenson, anche se con un profilo nettamente più moderno. Infine il Centaur, dapprima inciso privatamente dalla American Type Founders e quindi commercializzato dalla Monotype, risale a qualche anno dopo. Anche in questo caso Bruce Rogers fece ricorso alle fotografie, ma tracciandovi sulla superficie dei tratti con penna piatta diede al suo carattere un'impronta molto più calligrafica. Le diverse tecniche analizzate da Zapf fanno intendere la problematicità delle scelte progettuali messe in campo all'epoca, così come del resto accade ancora oggi.

Per il lavoro al Subiaco digitale, oltre a documenti disponibili in rete, mi sono avvalso delle fotografie di alta qualità fornitemi gentilmente da Riccardo Olocco, che ritraggono frammenti delle *Opere* di Lattanzio del 1465, conservate presso la Biblioteca Civica di Verona. Sebbene le tecniche digitali abbiano cambiato molto il modo di lavorare su supporti fotografici, l'adattamento necessario alla creazione di un carattere richiede approcci non dissimili da quelli del passato. Riprodurre esattamente quanto fatto secoli or sono è un'impresa disperata, sempre che quello sia il vero obiettivo. Un revival digitale non solo deve tener conto di aspetti estetici, ma anche essere rapportato alle tecnologie di uso corrente, per un impiego che sia al tempo stesso efficace e pertinente. Questi argomenti vengono affrontati da Paul Shaw nella discussione circa il trasporto in digitale di caratteri del passato (SHAW, 2017). Nel percorso storico che viene tracciato, sono presentati sia il Golden Type, digitalizzato dalla ITC, che il Doves Type, un faticoso recupero promosso da Robert Green e ultimato nel 2015. Un approfondimento sulle operazioni messe in campo per la riproduzione del Golden Type vengono discusse da Helga Jörgensen e Sigrid Engelmann, e alcune soluzioni intraprese

sono tenute presenti nelle operazioni di costruzione del Subiaco (KAROW, 1994). Ad esempio, la modalità in cui vengono gestite le intersezioni fra le diverse linee del contorno (*path*). Ritornando al codice della lettera "b", queste vengono risolte individuando il momento di incontro fra due percorsi piuttosto che le relative coordinate. Con METAFONT è possibile specificare tale punto, tenendo presente questo principio:

```
numeric t,u;
(t,u) = path_a intersectiontimes path_b;
```

Successivamente, sarà quindi possibile sottrarre ai rispettivi percorsi una piccola frazione di ognuno, con il parametro `blob`, e collegare così le porzioni rimanenti con una curva, eliminando di fatto lo spigolo dell'intersezione. In questo modo si può ottenere una transizione morbida da un tracciato all'altro, simulando da un lato le macchie d'inchiostro nel processo di stampa a torchio (*ink spread*) e favorendo dall'altro la conversione dell'immagine da vettoriale a bitmap per la stampa digitale:

```
subpath(0,t-blob) of path_a ...
subpath(u+blob,infinity) of path_b;
```

Riprendendo il discorso sul Subiaco della Ashendene Press, bisogna considerare che il carattere fu adattato per comporre testi in lingue moderne. Mi è sembrato adeguato perciò estendere l'alfabeto del Subiaco digitale, seguendo gli stessi criteri, anzi ampliandone le possibilità con diacritici e punteggiature diffuse nelle varie lingue europee. Per semplificare l'operazione, anche in questo caso sono ricorso a definizioni in grado di riprodurre la stessa figura in diversi glifi. Per esempio, nel caso dell'*accento acuto*, distribuito sulle vocali minuscole e maiuscole, si pone l'esigenza di collocare il diacritico in una posizione precisa, in base a un punto di riferimento riconducibile alla lettera stessa. Ne riporto il codice esplicativo:

```
vardef acute_acc(suffix $)(expr xacc,yacc) =
x$=xacc; x$'=x$+2.75hair;
y$=yacc-.125x_height; y$'=yacc+.125x_height;
numeric delta; delta:=angle(z$-z$');
penpos$(1.7thin,delta-90);
penpos$(3.7thin,delta-90);
acute$=z$'l---z$n..tension 2..z$m---z$'r
..tension 2..cycle;
penlabels($,$');
enddef;
path acute[];
```

Con il suffisso `$` viene creato un nuovo punto nello spazio, di cui a seguire vengono indicate le coordinate, `xacc` e `yacc`, in relazione ai parametri riferiti al glifo su cui poggia. Quindi viene generato un percorso ciclico tra due punti a esso correlati. L'istruzione per generare un *accento acuto* sommato a una vocale è pertanto la seguente:

```
acute_acc(15,2.25hair,acc_height);
fill acute15;
```

Canto Decimosecondo.
SI TOSTO come l'ultima parola
 La benedetta fiamma per dir tolse,
 A rotar cominciò la santa mola;
 E nel suo giro tutta non si volse
 Prima ch'un'altra di cerchio la chiuse,
 E moto a moto, e canto a canto colse;
 Canto, che tanto vince nostre Muse,
 Nostre Sirene, in quelle dolci tube,
 Quanto primo splendor quel ch'ei refuse.
 Come si volgon per tenera nube
 Due archi paralleli e concolori,
 Quando Junone a sua ancella iube,
 Nascendo di quel d'entro quel di fuori,
 A guisa del parlar di quella vaga,
 Ch'amor consunse come sol vapori;
 E fanno qui la gente esser presaga,
 Per lo patto che Dio con Noè pose,
 Del mondo che giammai più non si allaga:
 Così di quelle sempiterne rose
 Volgeansi circa noi le due ghirlande,
 E sì l'estrema all'ultima rispose.

82

Figura 4: Una pagina della *Commedia* di Dante, nello stile della Ashendene Press (KNIGHT, 2012).

Per generare un file OpenType dai sorgenti METAFONT sono ricorso a `mf2outline` di Linus Romer, uno script Python in grado di invocare dapprima METAPOST come sostituto di METAFONT per la sintesi dei contorni, quindi FontForge di George Williams per la compilazione del file `.otf`. Successivamente, sempre con FontForge, ho aggiunto un *feature file*, che rende possibile effettuare automaticamente la sostituzione dei glifi (ad esempio le legature) oltre a sporadiche correzioni di spaziatura fra caratteri. La struttura rigorosa della componente orizzontale dei glifi, come specificato in precedenza, basata sulla ripetizione ritmica di unità (*cadence units*, nel lessico di Blokland) è in grado di riprodurre con fedeltà la successione dei caratteri sia del Subiaco originale che della trasposizione della Ashendene. Le correzioni di crenatura, ridotte a poche coppie di glifi, sono riconducibili ai medesimi parametri.

Il risultato finale è una font digitale che, malgrado le numerose varianti, non dispone di tutti i glifi di un alfabeto moderno. Non è corredata di corsivo, grassetto, maiuscolo e via dicendo. Consente, però, di compilare dei testi in stile austero e comunque sufficientemente leggibili da risultare gradevoli. Come ho illustrato in questo scritto, il Subiaco è un tipo che già all'origine manifesta componenti ibride, che vengono mantenute in tutte le sue trasformazioni. Gli stessi approcci, nonché gli strumenti da me messi in opera per realizzarlo, sono eterogenei e multiformi, in una sinergia che può svelare il suo valore solo nei risultati ottenuti. Il fatto stesso che riviva oggi in questa trasposizione digitale ci fa capire quanto il Subiaco sia un classico, e come il modello archetipico travalichi il tempo e le situazioni.

Riferimenti bibliografici

- BLOKLAND, F.E. (2016). *On the origin of patterning in movable Latin type: Renaissance standardisation, systematisation, and unitisation of texture and roman type*. Tesi di Dottorato, Leiden University. <https://openaccess.leidenuniv.nl/handle/1887/43556>.
- BOARDLEY, J. (2019). *Typographic Firsts. Adventures in Early Printing*. Bodleian Library.
- BVMC (2019). «Cicero, De Oratore, p.159. Biblioteca Virtual Miguel de Cervantes». http://www.cervantesvirtual.com/obra-visor/de-oratore--0/html/01bdc2d6-82b2-11df-acc7-002185ce6064_166.html.
- CICERO, M.T. (2015). *De Oratore. Subiaco 1465. Ristampa anastatica*. Iter Edizioni.
- COCKERELL, S.C. (1898). *A note by William Morris on his aims in founding the Kelmscott Press*. Kelmscott Press. <https://archive.org/details/ANoteByWilliamMorrisOnHisAimsInFoundingTheKelmscottPressTogether>.
- DE GREGORI, L. (1942). «I tipi sublacensi». *Studi e ricerche sulla storia della stampa del Quattrocento*, pp. 47–61.
- FANTI, S. (2013). *Trattato di scrittura. Teorica et pratica de modo scribendi (Venezia 1514)*. Salerno Editrice.
- ICCU (2019). «ManusOnLine. Istituto Centrale per il Catalogo Unico». https://manus.iccu.sbn.it/opac_ElencoSchedeDiUnFondo.php?ID=145.
- KAROW, P. (1994). *Font Technology. Descriptions and Tools*. Springer-Verlag.
- KNIGHT, S. (2012). *Historical Types. From Gutenberg to Ashendene*. Oak Knoll Press.
- KNUTH, D.E. (1986). *The METAFONTbook*. Computers & typesetting. Addison-Wesley.
- MCNEIL, P. (2017). *The Visual History of Type*. Laurence King Publishing.
- SHAW, P. (2017). *Revival Type. Digital typefaces inspired by the past*. Yale University Press.
- TW (2019). «Type 1:120R. Typenrepertorium der Wiegendrucke. Staatsbibliothek zu Berlin». <https://tw.staatsbibliothek-berlin.de/ma10000>.
- VINCOLETTA, C. (2013). «Gli alfabeti romani di Francesco Griffo». *ArsTEXnica*, (16), pp. 52–57. <http://www.guitex.org/home/numero-16>.
- ZAPF, H. (1991). *Dalla calligrafia alla fotocomposizione*. Edizioni Valdomega.

▷ Claudio Vincoletto
Torino
claudio dot vincoletto at
gmail dot com